Contents

CONTEXT Search_C0

CONSTANTS

n Size of the array

f The function that models the array

v The value to locate in the array

AXIOMS

axm1: $n > 0$

The array does not have a zero length

axm2: $v \in \mathbb{N}$

axm3: $f \in 1 \mathinner{..} n \to \mathbb{N}$

A total function from indexes to naturals

axm4: $v \in ran(f)$

The value we are looking for is in the array

END

MACHINE Search_M0

This version checks array positions randomly

SEES Search_C0

VARIABLES

r

INVARIANTS

inv1: $r \in dom(f)$

r always within the boundaries of the array

EVENTS

Initialisation

begin

act1: $r :\in dom(f)$

end

Event Progress ⟨anticipated⟩ $\widehat{=}$

Did not find the value.

"Anticipated" means that a refinement of this event will be convergent.

when

grd1: $f(r) \neq v$

then

act1: $r :\in dom(f)$

end

Event Finish ⟨ordinary⟩ $\widehat{=}$

when

grd1: $f(r) = v$

Value found

then

skip

end

END

MACHINE Search_M1

Modeling linear seach

REFINES Search_M0

SEES Search_C0

VARIABLES

r

INVARIANTS

inv1: $v \in f[r \mathinner{..} n]$

To prove that we eventually find the element we are looking for

VARIANT

$n - r$ To prove formally that we terminate

EVENTS

Initialisation

begin

act1: $r := 1$

We start on one end of the array

end

Event Progress ⟨convergent⟩ $\widehat{=}$

"Convergent": we want to ensure it is eventually non-eligible

refines Progress

when

grd1: $f(r) \neq v$

If not found, v must be to the right of r (because of inv1)

then

act1: $r := r + 1$

Move one step forward

end

Event Finish ⟨ordinary⟩ $\widehat{=}$

extends Finish

when

grd1: $f(r) = v$

Value found

then

skip

end

END